#include "include\Common.fxh"

/*
% Blinn-Phong.
date: 05/29/12
*/
/*****************Un-Twealables**********************************/


//after colon is a semantic usefull later for knowing what variables this shader has
float4x4 WorldViewProjection : WORLDVIEWPROJECTION <string UIWidget="None";>;// < = annoation only for FXCOMPOSER 
float4x4 WorldInverseTranspose : WORLDINVERSETRANSPOSE <string UIWidget="None";>;
float4x4 World:WORLD <string UIWidget = "None";>;
float4x4 ViewInverse:VIEWINVERSE <string UIWidget = "None";>;

/****************Tweakables************************************/
/**Texture**/
texture ColorTexture : COLOR 
< string ResouceName = "default_color.dds";
  string UIName = "Color Texture";
  string ResourceType = "2D";>;
  
/**Ambient**/
float4 AmbientColor : COLOR
< string UIName = "Ambient Light";
  string UIWidget = "Color";
> = {1.0f,1.0f,1.0f,0.0f};

/**Diffuse**/
float3 LightDirection : DIRECTION 
< string Object = "DirectionalLight0";
  string UIName = "Light Direction";
  string Space = "World";
> = {0.0f , 0.0f, -1.0f};
float4 LightColor : DIFFUSE
< string UIName = "Light Color";
  string UIWidget = "Color";
> = {1.0f, 1.0f , 1.0f , 1.0f};

/**FOG**/
float3 FogColor
< string UIName = "Fog Color";
  string UIWidget = "Color";
> = {0.5f,0.5f,0.5f};
float FogStart = {20.0f};
float FogEnd = {40.0f};
bool FogEnabled = true;

/**PHONG**/
float4 SpecularColor : SPECULAR < 
	string UIName = "Specular Color";
	string UIWidget = "Color";
> = {1.0f,1.0f,1.0f,1.0f};

float SpecularPower : SPECULARPOWER <
	string UIName = "Specular Power";
	string UIWidget = "slider";
	float UIMin = 1.0;
	float UIMax = 255.0;
	float UIStep = 1.0;
> = {8.0f};

/***************Un-Tweakable************************************/
sampler2D ColorSampler = sampler_state
{	
	Texture = <ColorTexture>; //the texture it will be grabbing pixel date from
	Magfilter = LINEAR;		  //zoom in - does linear interpelation
	Minfilter = LINEAR;		  //zoom out
	Mipfilter = LINEAR;
	AddressU = Wrap;		  //wrap or clamp
	AddressV = Wrap;		  //uv values or 0 to 1 oustide that they must either be wrapped(round up to the next) or clamped(made to fit in 0to1)
};

/*****************Data Structures****************************/
struct VS_INPUT
{
	float4 ObjectPosition: POSITION0;
	float2 TextureCoordinate : TEXCOORD0;
	float4 Normal:NORMAL0;
};

struct VS_OUTPUT
{
	float4 ScreenPosition: POSITION0;
	float2 TextureCoordinate: TEXCOORD0;
	float3 Normal : TEXCOORD1;	//no noraml symantic for outup of texture shader...not important simply storing a 4 float
	float3 LightDirection: TEXCOORD2;
	float  FogAmount : TEXCOORD3;
	float4 ViewDirection: TEXCOORD4; //from the surface to the camera
};

struct PS_OUTPUT
{
	float4 Color:COLOR0;
};

/*****************Vertex Shader*****************************/
VS_OUTPUT vertex_shader(VS_INPUT IN)
{
	VS_OUTPUT OUT = (VS_OUTPUT)0; //memset clear garbage
	OUT.ScreenPosition = mul(IN.ObjectPosition, WorldViewProjection);
	OUT.TextureCoordinate = get_corrected_texture_coordinate(IN.TextureCoordinate);
	OUT.Normal = normalize(mul(IN.Normal, WorldInverseTranspose)); //place in the correct coordinate space
	OUT.LightDirection = normalize(-LightDirection); //want direction from vertex to light
	
	//FOG
	float4 worldPosition = mul(IN.ObjectPosition,World);
	float eyePosition = ViewInverse[3].xyz;
	OUT.FogAmount = get_fog_amount(eyePosition, worldPosition, FogStart, FogEnd);
	
	//PHONG
	OUT.ViewDirection = normalize(eyePosition - worldPosition);
	return OUT;
}

/*****************Pixel Shader*******************************/
PS_OUTPUT pixel_shader(VS_OUTPUT IN)
{
	PS_OUTPUT OUT = (PS_OUTPUT)0;
	
	float3 normal = normalize(IN.Normal);
	float3 lightDirection = normalize(IN.LightDirection);
	float3 viewDirection = normalize(IN.ViewDirection);
	float4 color = tex2D(ColorSampler, IN.TextureCoordinate);
	float4 specular = 0;
	float n_dot_l = dot(lightDirection, normal);
	float3 halfVector = normalize(lightDirection + viewDirection);
	
	if(n_dot_l > 0)
	{
		specular = min(pow(saturate(dot(normal,halfVector)), SpecularPower),color.w);
		specular = get_color_contribution(SpecularColor,specular);
	}
	
	float4 ambient = get_color_contribution(AmbientColor, color);
	float4 diffuse = get_color_contribution(LightColor,saturate(dot(lightDirection,normal))) * color;
	
	OUT.Color = ambient + diffuse + specular;//more diffuses would be added here
	if(FogEnabled)
	{
		OUT.Color.rgb = apply_fog(FogColor, OUT.Color, IN.FogAmount);
	}
	return OUT;
}

/****************Techniques**********************************/
technique main
{
	pass p0 
	{
		VertexShader = compile vs_3_0 vertex_shader();
		PixelShader = compile ps_3_0 pixel_shader();
	}
}
